

فصل دهم

تعریف ماکرو (MACRO) و کاربرد آن

تعریف ماکرو:

هر ماکرو با شبه دستور MACRO در ابتدای برنامه و قبل از تعریف سگمنت، به طریق زیر تعریف می شود.

Comment ; [نام پارامترها] MACRO نام ماکرو

دستورات اسمبلی

ENDM

برای تعریف ماکرو:

1- در شبه دستور ماکرو یک نام اختیاری برای ماکرو انتخاب می شود و در مقابل آن، نام پارامترها (اختیاری است) که در حقیقت نام متغیرهایی هستند که در دستورات بدنه ماکرو استفاده می شوند، قرار می گیرد. علاوه بر این پارامترهای ماکرو ثباتهای CPU نیز می توانند باشند.

2- در بدنه ماکرو تعداد دستور و توضیحات مربوطه قرار داده می شود.

3- در انتها شبه دستور ENDM پایان ماکرو را اعلام می نماید.

در صورتی که برنامه اصلی اسمبلی نیاز به دستورات بدنه ماکرو باشد، در این صورت فقط ذکر نام ماکرو یا دستور ماکرو، کافیت، البته در صورت لزوم، با نام ماکرو، متغیری که به جای پارامتر ماکرو، باید قرار داده شود، نیز ذکر می گردد. برای روشن شدن مطلب مثالهای زیر را بررسی می کنیم.

مثال

در برنامه‌های اسمبلی، پاک کردن مانیتور خیلی استفاده می‌شود، لذا می‌توان یک ماکرو با استفاده از سرویس 06H دستور INT 10H طبق زیر تعریف نمود.

```

;      Define MACRO
;      -----
CLESCR MACRO      ;Begin MACRO
;This MACRO clear monitor
;Using service 6 INT 10H
      MOV AX,0600H ;1-Service 6H
      MOV BH,07    ;2-White black
      MOV CX,0     ;3-Left up
      MOV DX,184FH ;4-Right down
      INT 10H      ;5-Interrupt
      ENDM        ;End of MACRO

```

بدنه ماکرو

ماکرو فوق نام ماکرو CLESCR است، و مانند برنامه‌های اسمبلی، توضیحات در قسمت‌های مختلف بدنه ماکرو نوشته شده است.

اگر در سگمنت کُد برنامه اسمبلی دستور ماکرو CLESCR به کار برده شود (مانند دستور ۱۶ برنامه کس ۱۷-۱) مانیتور پاک می‌گردد.

مثال

تعریف ماکروی پارامتر دار: فرض می‌کنیم می‌خواهیم با سرویس 09H دستور INT 21H، پیغام‌های customer name is و customer address is را روی مانیتور بنویسیم.

بن صورت ماکرو زیر را با نام PEYGHAM و با پارامتر STRING1 طبق زیر تعریف می‌نمائیم.

```

;
PEYGHAM MACRO STRING1 ;Begin MACRO
;This MACRO display on monitor
;DX=Address , Service 9H INT 21H
      MOV AH,09H    ;6-Service 9H
      LEA DX,STRING1 ;7-DX=Offset
      INT 21H      ;8-Interrupt
      ENDM        ;End of MACRO

```

بدنه ماکرو

ملاحظه می‌شود، پارامتر STRING1 در بدنه ماکرو در دستورات استفاده شده است. حال سگمنت داده برنامه اصلی پیغام‌های زیر را تعریف نمائیم:

```

MESSAG1 DB 'customer name is','$'
MESSAG2 DB 'customer address is','$'

```

اگر در سگمنت کُد برنامه اسمبلی دستور ماکرو PEYGHAM با پارامتر MESSAG1 به

PEYGHAM MESSAG1

باشد، پارامتر MESSAG1 در دستور ماکروی فوق، جانشین پارامتر STRING1 در تعریف ماکرو

می‌شود. یعنی برنامه اسمبلر در دستور LEA بدنه ماکرو مقدار MESSAGE را در رجیستر ESI قرار می‌گذارد، لذا MESSAGE که در سگمنت داده تعریف شده یعنی customer name is نوشته می‌شود.
 و اگر در سگمنت کُد برنامه، دستور ماکرو:
 CUSTOMER MESSAGE
 را بنویسیم در دستور LEA بدنه ماکرو، مقدار MESSAGE به جای STRING1 قرار می‌گیرد. در سگمنت MESSAGE که در سگمنت داده تعریف شده یعنی customer address is نوشته می‌شود. در ضمن ماکروهای فوق در برنامه شکل (۱۷-۱) استفاده شده است.

مثال

تعریف ماکرو برای بردن مکان‌نما به سطر و ستون مورد نظر: برای این منظور سرویس 02H دستور INT 10H استفاده می‌نمائیم و ماکرو CURSOR را با پارامترهای سطر ROW، و ستون COLUMN طبق زیر تعریف می‌کنیم.

```

;
CURSOR MACRO ROW,COLUMN ;Begin
;This MACRO sets cursor location
;ROW,COLUMN, INT 10 ,service 2
    MOV AH,02      ;9-Service 2
    MOV BH,00      ;10-Page 00
    MOV DH,ROW     ;11-Row
    MOV DL,COLUMN  ;12-Column
    INT 10H        ;13-Interrupt
ENDM                ;End of MACRO
  
```

در ماکروی فوق، پارامترهای سطر ROW و ستون COLUMN در دستورات بدنه ماکرو استفاده شده‌اند. اگر در سگمنت کُد برنامه اصلی دستور ماکرو CURSOR 10, 40 را بنویسیم، (برنامه شکل (۱۷-۱)، مقدار 40 و 10 به ترتیب از سمت چپ جایگزین COLUMN و ROW دستورات بدنه ماکرو می‌شوند و با اجرای ماکرو مذکور، مکان‌نما به سطر 10 و ستون 40 می‌رود. و یا اگر در برنامه مذکور دستور ماکرو CURSOR 20, 40 نوشته شود، مقدار 40 و 20 به ترتیب جایگزین COLUMN و ROW دستورات بدنه ماکرو می‌شوند، و با اجرای ماکروی مذکور، مکان‌نما به سطر 20 و ستون 40 می‌رود.
 با توجه به مثال‌های فوق ملاحظه می‌شود که، می‌توان ماکرو را چندین بار در سگمنت کُد برنامه اصلی اسمبلی استفاده نمود، مثال‌های بعدی این موضوع را در عمل نشان می‌دهد.

مثال

برنامه‌ای با سه ماکرو PEYGHAM، CLESCR و CURSOR بنویسید که:
 ۱- مانیتور را پاک کند.
 ۲- مکان‌نما را به سطر 10 و ستون 40 ببرد.
 ۳- پیغام customer name is را روی مانیتور بنویسد.
 ۴- مکان‌نما را به سطر 20 و ستون 40 ببرد.
 ۵- پیغام customer address is را روی مانیتور بنویسد.

حل: برنامه خواسته شده مطابق شکل (۱۷-۱) می‌باشد، که در این برنامه ماکروهای CLESCR، PEYGHAM و CURSOR، به ترتیب جهت پاک کردن مانیتور، نوشتن پیغام روی مانیتور و تغییر محل مکان‌نما، در ابتدای برنامه مذکور تعریف شده‌اند و سپس سگمنت‌های پشته، داده و کد قرار گرفته‌اند. در برنامه مذکور:

(۱) ماکرو CLESCR MACRO: برای پاک کردن مانیتور با استفاده از سرویس 6 دستور INT 10H تعریف شده است.

● دستورات ۱ و ۲: شماره سرویس وقفه، و رنگ مانیتور را مشخص می‌نمایند.

● دستورات ۳ و ۴: مختصات گوشه‌های چپ بالا و راست پایین را تعیین می‌کنند.

```

PAGE 110,100
TITLE 'micro_1.asm' example MICRO
;-----
;
;           Define MACRO
;-----
CLESCR MACRO                ;Begin MACRO
;This MACRO clear monitor
;Using service 6 INT 10H
    MOV AX,0600H           ;1-Service 6H
    MOV BH,07              ;2-White black
    MOV CX,0               ;3-Left up
    MOV DX,184FH           ;4-Right down
    INT 10H                ;5-Interrupt
    ENDM                   ;End of MACRO
;
PEYGHAM MACRO STRING1 ;Begin MACRO
;This MACRO display on monitor
;DX=Address , service 9H INT 21H
    MOV AH,09H             ;6-Service 9H
    LEA DX,STRING1        ;7-DX=Offset
    INT 21H                ;8-Interrupt
    ENDM                   ;End of MACRO
;
CURSOR MACRO ROW,COLUMN ;Begin
;This MACRO sets cursor location
;ROW,COLUMN, INT 10 ,service 2
    MOV AH,02              ;9-Service 2
    MOV BH,00              ;10-Page 00
    MOV DH,ROW             ;11-Row
    MOV DL,COLUMN          ;12-Column
    INT 10H                ;13-Interrupt
    ENDM                   ;End of MACRO
;-----

```

تعریف ماکرو ۱

تعریف ماکرو ۲

تعریف ماکرو ۳

```

DATASG SEGMENT 'DATA'
MESSAG1 DB 'customer name is','$'
MESSAG2 DB 'customer address is','$'
DATASG ENDS
} سگمنت داده

-----

;      3- Define code segment
;
CODESG SEGMENT 'CODE'
ASSUME SS:STACKSG,DS:DATASG,CS:CODESG
MAIN   PROC FAR
MOV AX,DATASG ;14-Initial
MOV DS,AX     ;15- DS
;
      CLESCR      ;16-CLESCR
      CURSOR 10,40 ;17-CURSOR
      PEYGHAM MESSAG1 ;18-PEYGHAM
      CURSOR 20,40 ;19-CURSOR
      PEYGHAM MESSAG2 ;20-PEYGHAM
EXIT:  MOV AX,4C00H ;21-End of
      INT 21H     ;22- processing
MAIN   ENDP
CODESG ENDS
END MAIN
} سگمنت کد

```

شکل (۱۷-۱) نمونه‌ای از برنامه اسمبلی با استفاده از ماکرو

- دستور ۵: وقفه را فعال می‌نماید، تا مانیتور پاک گردد.
- و بالاخره شبه دستور ENDM پایان ماکرو را اعلام می‌دارد.
- ماکرو PEYGHAM MACRO STRING1: برای نوشتن یک پیغام روی مانیتور، با پارامتر STRING1 است که از سرویس 09H دستور INT 21H استفاده شده است.
- با دستور ۶: شماره سرویس در ثبات AH قرار می‌گیرد. و با دستور ۷، افسس آدرس STRING1 در ثبات DX گذارده می‌شود. دستور ۸ وقفه 21H را فعال می‌کند.
- پایان ماکرو نیز با شبه دستور ENDM اعلام می‌گردد.

- ماکرو `COLUMN` , `CURSOR MACRO ROW` : برای تغییر محل مکان‌نما، به سطر `ROW` و ستون `COLUMN` یا استفاده از سرویس `02H` دستور `INT 10H` می‌باشد.
 - دستور ۹ : شماره سرویس `02H` را در ثبات `AH` قرار می‌دهد. دستور ۱۰ صفحه صفر مانی‌تور را فعال می‌کند. دستور ۱۱، مقدار سطر `ROW` را در ثبات `DH` قرار می‌دهد و دستور ۱۲ مقدار ستون `COLUMN` را در ثبات `DL` می‌گذارد و دستور ۱۳ وقفه `10H` را فعال می‌نماید.
 - پایان ماکرو نیز با شبه دستور `ENDM` اعلام می‌گردد.
 - سپس سگمنت‌های برنامه به طریق زیر تعریف می‌شوند:
 - برای سگمنت پشته `STACKSG` مانند برنامه‌های گذشته `32H` کلمه رزرو شده است.
 - در سگمنت داده `DATASG`، متغیرهای `MESSAG1` و `MESSAG2` به ترتیب برای نوشتن `customer name is` و `customer address is` تعریف شده‌اند.
 - در سگمنت کد `CODESG`:
 - دستورات ۱۴ و ۱۵ : مقدار اولیه به ثبات سگمنت داده `DS` می‌دهند.
 - دستور ۱۶ : ماکرو `CLESCR` را فرا می‌خواند، که مانی‌تور را پاک کند.
 - دستور ۱۷ : ماکرو `CURSOR 10, 40` را فرا می‌خواند که مکان‌نما را به سطر ۱۰ و ستون ۴۰ ببرد.
 - دستور ۱۸ : ماکرو `PEYGHAM MESSAG1` با پارامتر `MESSAG1` را فرا می‌خواند، که پیغام `MESSAG1`، که در سگمنت داده تعریف شده، یعنی `customer name is` در محل جاری مکان‌نما نوشته شود.
 - دستور ۱۹ : ماکرو `CURSOR 20, 40` با پارامتر ۲۰ و ۴۰ را فرا می‌خواند، که مکان‌نما را به سطر ۲۰ و ستون ۴۰ ببرد.
 - دستور ۲۰ : ماکرو `PEYGHAM MESSAG2` با پارامتر `MESSAG2` را فرا می‌خواند که پیغام `MESSAG2`، که در سگمنت داده تعریف شده، یعنی `customer address is` در محل جدید مکان‌نما نوشته شود.
 - و بالاخره دستورات ۲۱ و ۲۲ : برگشت به سیستم عامل است.
- برنامه مذکور پس از اسمبل و پیوند دادن، اجرا گردیده، که در نتیجه پیغام‌های زیر روی مانی‌تور نوشته شده‌اند.

```
customer name is
customer address is
```